

NAG Library Routine Document

D01AUF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

D01AUF is an adaptive integrator, especially suited to oscillating, nonsingular integrands, which calculates an approximation to the integral of a function $f(x)$ over a finite interval $[a, b]$:

$$I = \int_a^b f(x) dx.$$

2 Specification

```

SUBROUTINE D01AUF (F, A, B, KEY, EPSABS, EPSREL, RESULT, ABSERR, W, LW,      &
                  IW, LIW, IFAIL)
INTEGER           KEY, LW, IW(LIW), LIW, IFAIL
REAL (KIND=nag_wp) A, B, EPSABS, EPSREL, RESULT, ABSERR, W(LW)
EXTERNAL         F

```

3 Description

D01AUF is based on the QUADPACK routine QAG (see Piessens *et al.* (1983)). It is an adaptive routine, offering a choice of six Gauss–Kronrod rules. A global acceptance criterion (as defined by Malcolm and Simpson (1976)) is used. The local error estimation is described in Piessens *et al.* (1983).

Because D01AUF is based on integration rules of high order, it is especially suitable for nonsingular oscillating integrands.

D01AUF requires a subroutine to evaluate the integrand at an array of different points and is therefore amenable to parallel execution. Otherwise this algorithm with $KEY = 6$ is identical to that used by D01AKF.

4 References

Malcolm M A and Simpson R B (1976) Local versus global strategies for adaptive quadrature *ACM Trans. Math. Software* **1** 129–146

Piessens R (1973) An algorithm for automatic integration *Angew. Inf.* **15** 399–401

Piessens R, de Doncker–Kapenga E, Überhuber C and Kahaner D (1983) *QUADPACK, A Subroutine Package for Automatic Integration* Springer–Verlag

5 Arguments

- 1: F – SUBROUTINE, supplied by the user. *External Procedure*
 F must return the values of the integrand f at a set of points.

The specification of F is:

```

SUBROUTINE F (X, FV, N)
INTEGER           N
REAL (KIND=nag_wp) X(N), FV(N)

```

| | | |
|----|--|---------------|
| 1: | X(N) – REAL (KIND=nag_wp) array <i>On entry:</i> the points at which the integrand f must be evaluated. | <i>Input</i> |
| 2: | FV(N) – REAL (KIND=nag_wp) array <i>On exit:</i> FV(j) must contain the value of f at the point X(j), for $j = 1, 2, \dots, N$. | <i>Output</i> |
| 3: | N – INTEGER <i>On entry:</i> the number of points at which the integrand is to be evaluated. The actual value of N is equal to the number of points in the Kronrod rule (see specification of KEY). | <i>Input</i> |

F must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub) program from which D01AUF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 2: A – REAL (KIND=nag_wp) *Input*
On entry: a , the lower limit of integration.
- 3: B – REAL (KIND=nag_wp) *Input*
On entry: b , the upper limit of integration. It is not necessary that $a < b$.
- 4: KEY – INTEGER *Input*
On entry: indicates which integration rule is to be used.
KEY = 1
 For the Gauss 7-point and Kronrod 15-point rule.
KEY = 2
 For the Gauss 10-point and Kronrod 21-point rule.
KEY = 3
 For the Gauss 15-point and Kronrod 31-point rule.
KEY = 4
 For the Gauss 20-point and Kronrod 41-point rule.
KEY = 5
 For the Gauss 25-point and Kronrod 51-point rule.
KEY = 6
 For the Gauss 30-point and Kronrod 61-point rule.
Suggested value: KEY = 6.
Constraint: KEY = 1, 2, 3, 4, 5 or 6.
- 5: EPSABS – REAL (KIND=nag_wp) *Input*
On entry: the absolute accuracy required. If EPSABS is negative, the absolute value is used. See Section 7.
- 6: EPSREL – REAL (KIND=nag_wp) *Input*
On entry: the relative accuracy required. If EPSREL is negative, the absolute value is used. See Section 7.
- 7: RESULT – REAL (KIND=nag_wp) *Output*
On exit: the approximation to the integral I .

- 8: ABSERR – REAL (KIND=nag_wp) Output
On exit: an estimate of the modulus of the absolute error, which should be an upper bound for $|I - \text{RESULT}|$.
- 9: W(LW) – REAL (KIND=nag_wp) array Output
On exit: details of the computation see Section 9 for more information.
- 10: LW – INTEGER Input
On entry: the dimension of the array W as declared in the (sub)program from which D01AUF is called. The value of LW (together with that of LIW) imposes a bound on the number of sub-intervals into which the interval of integration may be divided by the routine. The number of sub-intervals cannot exceed LW/4. The more difficult the integrand, the larger LW should be.
Suggested value: LW = 800 to 2000 is adequate for most problems.
Constraint: LW \geq 4.
- 11: IW(LIW) – INTEGER array Output
On exit: IW(1) contains the actual number of sub-intervals used. The rest of the array is used as workspace.
- 12: LIW – INTEGER Input
On entry: the dimension of the array IW as declared in the (sub)program from which D01AUF is called.
 The number of sub-intervals into which the interval of integration may be divided cannot exceed LIW.
Suggested value: LIW = LW/4.
Constraint: LIW \geq 1.
- 13: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if IFAIL \neq 0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: D01AUF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

The maximum number of subdivisions allowed with the given workspace has been reached without the accuracy requirements being achieved. Look at the integrand in order to determine the integration difficulties. If necessary, another integrator, which is designed for handling the

type of difficulty involved, must be used. Alternatively, consider relaxing the accuracy requirements specified by EPSABS and EPSREL, or increasing the amount of workspace.

IFAIL = 2

Round-off error prevents the requested tolerance from being achieved. Consider requesting less accuracy.

IFAIL = 3

Extremely bad local integrand behaviour causes a very strong subdivision around one (or more) points of the interval. The same advice applies as in the case of IFAIL = 1.

IFAIL = 4

On entry, KEY \neq 1, 2, 3, 4, 5 or 6.

IFAIL = 5

On entry, LW < 4,
or LIW < 1.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

D01AUF cannot guarantee, but in practice usually achieves, the following accuracy:

$$|I - \text{RESULT}| \leq \text{tol},$$

where

$$\text{tol} = \max\{|\text{EPSABS}|, |\text{EPSREL}| \times |I|\},$$

and EPSABS and EPSREL are user-specified absolute and relative error tolerances. Moreover, it returns the quantity ABSERR which, in normal circumstances, satisfies

$$|I - \text{RESULT}| \leq \text{ABSERR} \leq \text{tol}.$$

8 Parallelism and Performance

D01AUF is not threaded in any implementation.

9 Further Comments

If IFAIL \neq 0 on exit, then you may wish to examine the contents of the array W, which contains the end points of the sub-intervals used by D01AUF along with the integral contributions and error estimates over these sub-intervals.

Specifically, for $i = 1, 2, \dots, n$, let r_i denote the approximation to the value of the integral over the sub-interval $[a_i, b_i]$ in the partition of $[a, b]$ and e_i be the corresponding absolute error estimate. Then, $\int_{a_i}^{b_i} f(x) dx \simeq r_i$ and $\text{RESULT} = \sum_{i=1}^n r_i$. The value of n is returned in $\text{IW}(1)$, and the values a_i, b_i, e_i and r_i are stored consecutively in the array W , that is:

$$\begin{aligned} a_i &= W(i), \\ b_i &= W(n+i), \\ e_i &= W(2n+i) \text{ and} \\ r_i &= W(3n+i). \end{aligned}$$

10 Example

This example computes

$$\int_0^{2\pi} x \sin(30x) \cos x \, dx.$$

10.1 Program Text

```
! D01AUF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d01aufe_mod

! D01AUF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
Implicit None
! .. Accessibility Statements ..
Private
Public :: f
! .. Parameters ..
Integer, Parameter, Public :: lw = 800, nout = 6
Integer, Parameter, Public :: liw = lw/4
Contains
Subroutine f(x,fv,n)

! .. Scalar Arguments ..
Integer, Intent (In) :: n
! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: fv(n)
Real (Kind=nag_wp), Intent (In) :: x(n)
! .. Intrinsic Procedures ..
Intrinsic :: cos, sin
! .. Executable Statements ..
fv(1:n) = x(1:n)*(sin(30.0E0_nag_wp*x(1:n)))*cos(x(1:n))

Return

End Subroutine f
End Module d01aufe_mod
Program d01aufe

! D01AUF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d01auf, nag_wp, x01aaf
Use d01aufe_mod, Only: f, liw, lw, nout
! .. Implicit None Statement ..
Implicit None
! .. Local Scalars ..
```

```

      Real (Kind=nag_wp)           :: a, abserr, b, epsabs, epsrel, pi,    &
                                result
      Integer                     :: ifail, key
! .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: w(:)
      Integer, Allocatable          :: iw(:)
! .. Executable Statements ..
      Write (nout,*) 'D01AUF Example Program Results'

      Allocate (w(lw),iw(liw))

      pi = x01aaf(pi)
      epsabs = 0.0E0_nag_wp
      epsrel = 1.0E-03_nag_wp
      a = 0.0E0_nag_wp
      b = 2.0E0_nag_wp*pi
      key = 6

      ifail = -1
      Call d01auf(f,a,b,key,epsabs,epsrel,result,abserr,w,lw,iw,liw,ifail)

      If (ifail>=0) Then
        Write (nout,*)
        Write (nout,99999) 'A      ', 'lower limit of integration', a
        Write (nout,99999) 'B      ', 'upper limit of integration', b
        Write (nout,99998) 'EPSABS', 'absolute accuracy requested', epsabs
        Write (nout,99998) 'EPSREL', 'relative accuracy requested', epsrel
      End If
      If (ifail>=0 .And. ifail<=3) Then
        Write (nout,*)
        Write (nout,99997) 'RESULT', 'approximation to the integral', result
        Write (nout,99998) 'ABSERR', 'estimate of the absolute error', abserr
        Write (nout,99996) 'IW(1) ', 'number of subintervals used', iw(1)
      End If

99999 Format (1X,A6,' - ',A30,' = ',F10.4)
99998 Format (1X,A6,' - ',A30,' = ',E9.2)
99997 Format (1X,A6,' - ',A30,' = ',F9.5)
99996 Format (1X,A6,' - ',A30,' = ',I4)
      End Program d01aufe

```

10.2 Program Data

None.

10.3 Program Results

D01AUF Example Program Results

```

A      -   lower limit of integration =   0.0000
B      -   upper limit of integration =   6.2832
EPSABS -   absolute accuracy requested =  0.00E+00
EPSREL -   relative accuracy requested =  0.10E-02

RESULT -   approximation to the integral = -0.20967
ABSERR -   estimate of the absolute error =  0.45E-13
IW(1)  -   number of subintervals used =    4

```
